

Teil 3





Abb. 1.3-6 Spannungsbedarf der halben Anordnung

Die vorstehende Betrachtung kann natürlich nur das Prinzip der Modellierung aufzeigen. Für eine Anwendung zur Untersuchung des realen Vorgangs müssen wesentlich mehr Abhängigkeiten (Nichtlinearitäten der Materialeigenschaften, Schmelzwärme des Materials, Wärmeabgabe durch Strahlung, Kontaktwiderstände) berücksichtigt werden. Hat man dann aber ein die Realität hinreichend widerspiegelndes Modell, kann eine Verfahrensdatierung und eine Fehlerursachenermittlung am Rechner oft besser als im Experiment erfolgen.

1.4 Elektroerosion

Fußpunktverdampfung

Bei der Elektroerosion bildet sich eine elektrische Entladung zwischen zwei Elektroden aus. An der Werkstückelektrode kommt es im Bereich des Fußpunktes dieser Entladung zu einer gewünschten Erwärmung bis zur Aufschmelzung und Verdampfung des Materials. Damit wird ein lokaler Abtrag bewirkt. An der Werkzeugelektrode ist dieser Vorgang unerwünscht. Er führt zum Verschleiß der Elektrode.

Die Auswirkungen des Energieeintrags an der Werkstückoberfläche sollen betrachtet werden. Der Kanal der sich ausbildenden elektrischen Entladung kann zylinderförmig angenommen werden. In dessen Fußpunkt wird ein Teil der Entladungsenergie in Wärme umgesetzt. Diese ergibt sich aus dem Katodenfall * Strom. Zielstellung der Betrachtung ist die Form des sich ausbildenden Kraters.

Im ersten Schritt soll eine grobe Orientierungsrechnung durchgeführt werden.



Teil 3



Am Programmanfang werden die Parameter vereinbart. Die Säule der Entladung beginnt mit einem kleinen Radius (radanf). Danach breitet sie sich mit der Radiuswachstumsgeschwindigkeit (radwa) für die Dauer des Stromwachstums aus. Der Strom beginnt bei Null und steigt linear mit dem Stromwachstum (strwa) bis zu seinem Endwert (strend). Für die Geometriegenerierung wird ein mittlerer Radius (radmi) vereinbart. Bis zu ihm soll relativ eng vernetzt werden. Der gesamte zu untersuchende Bereich soll bis zu maximalen Radius (radma) gehen.

```
!Funken auf eine Platte
!rotationssymmetrisch im Schnitt
!einfache Geometrie
!mit Enthalpie
              !Radius am Anfang
radanf=1e-6
strend=40
              !Strom-Ende
pulsd=2e-6
              !Puls-Dauer
              !Radius-Wachstum [um/us]
radwa=80
strwa=200e6
              !Strom-Wachstum
                                [A/s]
radmi=25e-6
              !Radius-mittlerer
radma=50e-6
              !Radius-Maximum
fupusp=14
              !Fußpunktspannungsabfall
```

```
/prep7
et,1,plane55,,,1
mp,kxx,1,84
mp,dens,1,7860
mptemp,1,300,1800,1810,2500,2800,3000
mpdata,enth,1,1,9.8e8,6.5e9,8.7e9,1.2e10,6.2e10,6.3e10
mpplot,enth,1
```



Abb. 1.4-1 Enthalpie (Steilheit verringert)





Teil 3

Diese Nichtlinearität kann zu Konvergenzproblemen der auszuführenden Berechnungen führen. Um eine orientierende Berechnung ausführen zu können, kann die folgende Vereinfachung des Enthalpieverlaufes vorgenommen werden.

```
mptemp,1,300,2800,5000
mpdata,enth,1,1,9.8e8,6.2e10,6.21e10
```

Die Geometrie des zu untersuchenden Bereichs wird durch zwei Viertel-Kreisflächen gebildet.

```
!-----Geometrie-----
pcirc,0,radmi,0,-90
pcirc,0,radma,0,-90
aovlap,all
!A1:innen A3:außen
```

Der innere Teil wird relativ eng (1µm) mit gleichmäßiger Maschenweite vernetzt.

```
esize,1e-6
mshape,0,2d
mshkey,1
amesh,1
```

Der äußere Teil wird grob vernetzt.

```
csys,1
lsel,s,loc,x,radmi,radma
lsel,u,loc,x,radmi
lsel,u,loc,x,radma
lesize,all,,,6,6
esize
amesh,3
```

Die Berechnung erfolgt transient. Die Temperatur wird als absolute aufgefasst. Deshalb wird zu Beginn alles auf 300 K gesetzt.

```
tunif,300
/solu
antype,trans
kbc,1
time,1e-12
nsubst,1
solve !Startwert
```

Die Berechnung des Endradius des Entladungskanals erfolgt aus den Vorgaben. Die Zeitschrittweite ergibt sich aus den Wachstumsgeschwindigkeiten. In den Anstieg müssen hinreichend viele Punkte gelegt werden. Damit ergibt sich die Anzahl der Berechnungsschritte.







Abb. 1.4-2 Vernetzte Fläche

```
radend=strend/strwa*radwa
sw=50e-9 !Schrittweite
schritte=pulsd/sw
```

Für jeden Berechnungsschritt wird der Strom und der aktuelle Fußpunktradius ermittelt.

```
*do,i,1,schritte
zeit=sw*i
!-----Strom-----
strom=strwa*zeit
*if,strom,gt,strend,then
strom=strend
*endif
!-----Radius-----
rad=radanf+radwa*zeit
*if,rad,gt,radend,then
rad=radend
*endif
```

Die Wärmestromdichte, die mit dem sf-Befehl übergeben wird, kann aus Strom, Fußpunktspannungsabfall und Fläche ermittelt werden.

!----Fläche----fla=3.1415*rad*rad





nsel,s,loc,y,0
nsel,r,loc,x,0,rad
sf,all,hflux,strom*fupusp/fla*1
nsel,all

Für jeden der Schritte werden Bedingungen vorgegeben.

```
time,zeit
autots,on !automatische Schrittweitensteuerung
neqit,50
nsubst,10,200,2
solve
*enddo
```

Die Auswertung als Animation erfolgt mit folgenden Anweisungen. Mit dem /contour-Befehl wird der Temperaturbereich zwischen 300 und 2900 K durch entsprechende Farben dargestellt. Bereiche mit einer Temperatur über 2900 K, die verdampftes Material enthalten, werden grau ausgewiesen.

/post1
/plopts,minm,0
/cont,,,300,200,2900
!/expand,27,axis,full,,10,1,polar,,180
!-----*do,i,1,schritte
set,i,last
plns,temp
/wait,0.2
*enddo



Abb. 1.4-3 Temperaturverteilung

Es entsteht ein Krater von ca. 30 µm Durchmesser und ca. 3 µm Tiefe.